# Babel, internationalization utilities
#

pkgname=python2-babel
pkgver=0.9.6
pkgrel=1
pkgdesc="A collection of tools for internationalizing Python applications"
url="http://babel.edgewall.org/"
license=('BSD')
arch=('ppc')
depends=('python2')
makedepends=('setuptools')			# python2-distribute
source=("http://pypi.python.org/packages/source/B/Babel/Babel-${pkgver}.tar.gz")
md5sums=('f0edcad03dfdb5505f337ef1a7690325')

build() {
	cd "${srcdir}/Babel-${pkgver}"

	# Patch
	for file in `find . -type f -name '*.py'`; do
		sed -i 's|env python|env python2.7|' ${file}
	done

	/usr/bin/python2.7 setup.py build
}

package() {
	cd "${srcdir}/Babel-${pkgver}"

	/usr/bin/python2.7 setup.py install --root=${pkgdir}

	install -D -m644 COPYING ${pkgdir}/usr/share/licenses/${pkgname}/COPYING
}
